'--------------------------------------------------------------


'


'                    M.v. Gersdorff


'--------------------------------------------------------------


'  file: PID_103_3.BAS


'  Temperaturregler mit PID-Verhalten


'  Verstärkungen über Pots am ADC einstellbar


'  Sollwert Wird vorgegeben


'  Ausgangsgröße ist ein PWM Signal für externen Power Mosfet


'--------------------------------------------------------------

Bild 1:


Obere Kurve ist die gemessene Temperatur des Lastwiderstandes. Die obere gestrichelte Linie ist der Sollwert (35 Grad).


Untere Kurve ist die mit einem RC Glied geglättete PWM-Spannung an dem Lastwiederstand. Das Andere Ende des Lastwiderstandes (22 Ohm 25 Watt)  ist an 13,8 Volt angeschlossen. Die Heizleistung ist also nach unten zunehmend. Die Schaltung wurde mit einem Atmega 103 Experimentierboard der Firma RIBU realisiert. Der Leistungswiderstand wurde mit einem IPS021L direkt aus dem Port angesteuert. Über ein angeschlossenen PC mit Terminal-Programm werden zur Kontrolle einige Größen am Bilsdschirm überwacht. 





LISTING BASCOM:





$regfile = "m103def.dat"                                    'Configuriert Den Atmega103





Config Adc = Single , Prescaler = Auto





Config Timer1 = Pwm , Pwm = 10 , Compare A Pwm = Clear Up , Compare B Pwm = Clear Up , Prescale = 1








'Verwendete Variablen


Dim Istwert As Single


Dim Sollwert As Single


Dim Integrator As Single


Dim Lastistwert As Single


Dim P As Single


Dim I As Single


Dim D As Single


Dim Delta As Single


Dim Ausgang As Single


Dim Pw As Word


Dim Wert As Word


Dim Kp As Single


Dim Kd As Single


Dim Ki As Single


Dim Zeit As Single


Dim Invertiert As Word


Dim Istwert_bit As Word





Start Adc














'Zeitinterwall für D und I-Anteil (muß, falls sehr klein,noch mit Rechenzeit korrigiert werden)


Zeit = 2











Lastistwert = Istwert                                       'Soll Sprung am Start verhindern





'Setzen der Anfangsbedingungen


Sollwert = 35                                               'Einheit ist Grad C


Integrator = 0











Do





   Istwert_bit = Getadc(0)


   Istwert = Istwert_bit * 0.0449658                        'Skalierung Verstärker mit Meßbereich 31+/- 23 Grad


   Istwert = Istwert + 7.9925


   Print "Temperatur=" ; Istwert








   'Verstärkungen des PID-Reglers                           Zur Optimierung sind die Verstärkungen über Pots  einstellbar


   Wert = Getadc(1)


   Kp = Wert / 10


   Print "kp= " ; Kp ; "(%PWM/°C)"





   Wert = Getadc(2)


   Kd = Wert


   Print "kd= " ; Kd ; "(%PWM/°C/sec)"





   Wert = Getadc(3)


   Ki = Wert / 100


   Print "ki= " ; Ki ; "(%PWM/°C*sec)"








   'Berechnung der Reglergrößen


      'Proportionalanteil


      Delta = Sollwert - Istwert


      P = Delta * Kp


      If P > 1023 Then P = 1023                             'Begrenzung der Ausgangsgröße


      Print "p=" ; P


	


      D = Lastistwert - Istwert


      D = D / Zeit


      D = D * Kd


      If D > 1023 Then D = 1023


      If D < -1023 Then D = -1023                           'Begrenzung der Ausgangsgröße


      Print "d=" ; D








      I = Delta * Zeit


      I = I * Ki


      If Pw <= 1 Then Goto Weiter


      If Pw < 1023 Then Integrator = Integrator + I


      Weiter:


      If Integrator < -1023 Then Integrator = -1023         'Begrenzung der Ausgangsgröße


      If Integrator > 1023 Then Integrator = 1023


      If Ki < 0.01 Then Integrator = 0


      Print "I=" ; I


      Print "Integrator=" ; Integrator





         'Berechnung der Ausgangsgrößen


         Ausgang = D


         Ausgang = Ausgang + P


         Ausgang = Ausgang + Integrator


         If Ausgang > 1023 Then Ausgang = 1023              'Begrenzung der Ausgangsgröße


         If Ausgang <= 0 Then Ausgang = 0


         Pw = Ausgang





         Print "PW= " ; Pw





         'Ansteuerung des Mosfets





         Compare1a = Pw








   'Zeitschritt und Rücksprung


    Lastistwert = Istwert





    Wait 2                                                  'Achtung diese Pause muß gleich lang wie "Zeit" sein





Loop























End


